NAG Library Routine Document

G13AUF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13AUF calculates the range (or standard deviation) and the mean for groups of successive time series values. It is intended for use in the construction of range-mean plots.

2 Specification

```
SUBROUTINE G13AUF (N, Z, M, NGRPS, RS, Y, MEAN, IFAIL)
INTEGER             N, M, NGRPS, IFAIL
REAL (KIND=nag_wp) Z(N), Y(NGRPS), MEAN(NGRPS)
CHARACTER(1)        RS
```

3 Description

Let $Z_1, Z_2, \ldots, Z_n$ denote n successive observations in a time series. The series may be divided into groups of m successive values and for each group the range or standard deviation (depending on a user-supplied option) and the mean are calculated. If n is not a multiple of m then groups of equal size m are found starting from the end of the series of observations provided, and any remaining observations at the start of the series are ignored. The number of groups used, k, is the integer part of n/m. If you wish to ensure that no observations are ignored then the number of observations, n, should be chosen so that n is divisible by m.

The mean, M_i, the range, R_i, and the standard deviation, S_i, for the ith group are defined as

$$M_i = \frac{1}{m}\sum_{j=1}^{m} Z_{l+m(i-1)+j}$$

$$R_i = \max_{1\le j\le m}\left\{Z_{l+m(i-1)+j}\right\} - \min_{1\le j\le m}\left\{Z_{l+m(i-1)+j}\right\}$$

and

$$S_i = \sqrt{\left(\frac{1}{m-1}\right)\sum_{j=1}^{m}\left(Z_{l+m(i-1)+j} - M_i\right)^2}$$

where $l = n - km$, the number of observations ignored.

For seasonal data it is recommended that m should be equal to the seasonal period. For non-seasonal data the recommended group size is 8.

A plot of range against mean or of standard deviation against mean is useful for finding a transformation of the series which makes the variance constant. If the plot appears random or the range (or standard deviation) seems to be constant irrespective of the mean level then this suggests that no transformation of the time series is called for. On the other hand an approximate linear relationship between range (or standard deviation) and mean would indicate that a log transformation is appropriate. Further details may be found in either Jenkins (1979) or McLeod (1982).

You have the choice of whether to use the range or the standard deviation as a measure of variability. If the group size is small they are both equally good but if the group size is fairly large (e.g., $m = 12$ for monthly data) then the range may not be as good an estimate of variability as the standard deviation.

4 References

Jenkins G M (1979) *Practical Experiences with Modelling and Forecasting Time Series* GJP Publications, Lancaster

McLeod G (1982) *Box–Jenkins in Practice. 1: Univariate Stochastic and Single Output Transfer Function/Noise Analysis* GJP Publications, Lancaster

5 Arguments

1: N – INTEGER *Input*

On entry: n, the number of observations in the time series.

Constraint: N $\geq$ M.

2: Z(N) – REAL (KIND=nag_wp) array *Input*

On entry: Z(t) must contain the tth observation Z_t, for $t = 1, 2, \ldots, n$.

3: M – INTEGER *Input*

On entry: m, the group size.

Constraint: M ≥ 2.

4: NGRPS – INTEGER *Input*

On entry: k, the number of groups.

Constraint: NGRPS = int(N/M).

5: RS – CHARACTER(1) *Input*

On entry: indicates whether ranges or standard deviations are to be calculated.

RS = 'R'
Ranges are calculated.

RS = 'S'
Standard deviations are calculated.

Constraint: RS = 'R' or 'S'.

6: Y(NGRPS) – REAL (KIND=nag_wp) array *Output*

On exit: Y(i) contains the range or standard deviation, as determined by RS, of the ith group of observations, for $i = 1, 2, \ldots, k$.

7: MEAN(NGRPS) – REAL (KIND=nag_wp) array *Output*

On exit: MEAN(i) contains the mean of the ith group of observations, for $i = 1, 2, \ldots, k$.

8: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N $<$ M,
or M < 2,
or NGRPS $\neq$ integer part of N/M.

IFAIL $= 2$

On entry, RS is not equal to `R' or `S'.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computations are believed to be stable.

8 Parallelism and Performance

G13AUF is not threaded in any implementation.

9 Further Comments

The time taken by G13AUF is approximately proportional to n.

10 Example

The following program produces the statistics for a range-mean plot for a series of 100 observations divided into groups of 8.

10.1 Program Text

```
    Program g13aufe

!     G13AUF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g13auf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
```

```
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, m, n, ngrps
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: mean(:), y(:), z(:)
!     .. Executable Statements ..
      Write (nout,*) 'G13AUF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Read in the problem size
      Read (nin,*) n, m

      ngrps = n/m

      Allocate (z(n),y(ngrps),mean(ngrps))

!     Read in data
      Read (nin,*) z(1:n)

!     Calculate summary statistics
      ifail = 0
      Call g13auf(n,z,m,ngrps,'RANGE',y,mean,ifail)

!     Display title
      Write (*,*) '     Mean      Range'
      Write (*,*) '   -------------------'
      Do i = 1, ngrps
        Write (nout,99999) mean(i), y(i)
      End Do

99999 Format (2(1X,F10.3))
    End Program g13aufe
```

10.2 Program Data

```
G13AUF Example Program Data
100 8                              : N,M
 101.0  82.0  66.0  35.0  31.0
   6.0  20.0  90.0 154.0 125.0
  85.0  68.0  38.0  23.0  10.0
  24.0  83.0 133.0 131.0 118.0
  90.0  67.0  60.0  47.0  41.0
  21.0  16.0   6.0   4.0   7.0
  14.0  34.0  45.0  43.0  49.0
  42.0  28.0  10.0   5.0   2.0
   0.0   1.0   3.0  12.0  14.0
  35.0  47.0  41.0  30.0  24.0
  16.0   7.0   4.0   2.0   8.0
  13.0  36.0  50.0  62.0  67.0
  72.0  48.0  29.0   8.0  13.0
  57.0 122.0 139.0 103.0  86.0
  63.0  37.0  26.0  11.0  15.0
  40.0  62.0  98.0 124.0  96.0
  65.0  64.0  54.0  39.0  21.0
   7.0   4.0  23.0  53.0  94.0
  96.0  77.0  59.0  44.0  47.0
  30.0  16.0   7.0  37.0  74.0   : End of X
```

10.3 Program Results

```
 G13AUF Example Program Results

      Mean      Range
    -------------------
     72.375    148.000
     70.000    123.000
```

```
43.500     84.000
29.750     45.000
 7.625     28.000
26.750     40.000
30.250     65.000
61.000    131.000
47.625     92.000
75.250     85.000
46.875     92.000
39.250     67.000
```